# FIJI (ImageJ) Macro to Automate Counting of *Perkinsus spp.*

Teresa Vaillancourt, Ph.D, SERC volunteer, December 2021

Accurate quantification of infection intensity may be obtained by the whole body burden assay. That assay is possible because of the remarkable ability of stainable Perkinsus cell walls to survive prolonged immersion in a 2M NaOH solution at 60° C, a solution that dissolves virtually all other cell and tissue components of the parasite and its host. Following processing, Perkinsus samples can be accurately counted using FIJI, an enhanced version of ImageJ.

The following macro is based on automatic separation of putative Perkinsus hypnospores and background by using the Phansalker local threshold algorithm. This macro may be used to analyze single images or an entire directory of images for whole body burden assays and for assays of filtered habitat water and/or feces.

What follows is an annotated version written in the ImageJ macro language; the actual commands are in **RED**. Adjacent images represent the working image resulting from the described segment of code.

```
/*
This macro is optimized for counting samples of RFTM
incubated, NaOH digested, Lugol's stained samples that have
been imaged with high resolution color micrographs.
Dimensional parameters are in "pixel" units.

The first steps record the name of the open image file (adjacent
example), and create copies for later use.
*/
FileName = getInfo("image.filename");
run("Duplicate...", "title=Copy1");
run("Duplicate...", "title=Copy2");
```



```
/*
The image is then split into RGB luminance channels. The
Green and Blue channels are discarded, and the Red
luminance channel (see adjacent image) becomes the working
image because it provides the best contrast separation
between the reddish background of Lugol's stained samples
and the blue-black Perkinsus hypnospores.
*/
selectWindow(FileName);
run("Split Channels");
close();
close();
```



/*
The working image is segmented into “background” and “target” pixels by using the Phansalkar local threshold algorithm. The result is a black and white image. Perkinsus hypnospores (as well as similarly sized debris) are reliably segregated when the algorithm’s radius variable is set to 100 pixels.



The watershed algorithm is used to automatically detect and draw a line between instances of clustered Perkinsus, although it is not 100% accurate – see later note regarding manual adjustment of final count.
*/

```
run("Auto Local Threshold", "method=Phansalkar radius=100 parameter_1=0 parameter_2=0
white");
run("Gaussian Blur...", "sigma=3");
run("Convert to Mask");
run("Watershed");
```

/*
Perkinsus hypnospores are circular or oval in cross-section, distinguishing them from irregularly shaped debris. Particles having diameters (in pixel units) in the expected range for Perkinsus hypnospores and having circular cross-sections are identified, enumerated, measured, tabulated, and masked using the “Analyze Particles” algorithm (adjacent image). Particles on image margins are excluded.



*/

```
run("Set Measurements...", "area perimeter fit shape display redirect=None decimal=2");
run("Analyze Particles...", "size=50-1000000 circularity=0.85-1.00 show=[Overlay Masks]
display exclude include summarize");
run("Labels...", "color=black font=12 show bold");
run("Magenta");
run("Flatten");
run("Copy");
close();
close();
```

/*
A montage of the original and “analyzed” image is created to simplify the process of authentication and correction of the automated count. In the “analyzed” portion of the montage, particles masked in **BLUE** were counted, those masked in **MAGENTA** were debris or possible Perkinsus hypnospores that did not meet the specified size and circularity criteria or that were clusters not separable by the “Watershed” algorithm. Upon inspection, the user may choose to

modify the count by adding specific Magenta-masked particles, e.g. the two associated with debris in the upper left quadrant, or deleting miscounted Blue-masked ones.
*/

```
selectWindow("Copy2");
run("Select All");
setBackgroundColor(0,0,0);
run("Clear", "slice");
run("Paste Control...");
run("Paste");
selectWindow("Copy1");
run("Images to Stack", "name=Stack title=[] use");
run("Make Montage...", "columns=2 rows=1 scale=1");
rename(FileName+"Montage");
```



The resulting tables, containing particle enumerations and shape information (pixel units), may be transferred to a spreadsheet and used for further analysis.

Results

File Edit Font Results

| | Label | Area | Perim. | Major | Minor | Angle | Circ. | AR | Round | Solidity |
|---|---|---|---|---|---|---|---|---|---|---|
| 1898 | IMG_0026.JPG (red) | 3172 | 210.65 | 63.82 | 63.28 | 28.39 | 0.90 | 1.01 | 0.99 | 0.97 |
| 1899 | IMG_0026.JPG (red) | 1815 | 159.20 | 49.22 | 46.95 | 32.77 | 0.90 | 1.05 | 0.95 | 0.97 |
| 1900 | IMG_0027.JPG (red) | 2944 | 202.99 | 61.75 | 60.71 | 132.64 | 0.90 | 1.02 | 0.98 | 0.97 |
| 1901 | IMG_0027.JPG (red) | 768 | 103.40 | 32.87 | 29.75 | 124.87 | 0.90 | 1.10 | 0.91 | 0.95 |
| 1902 | IMG_0027.JPG (red) | 1649 | 150.12 | 46.61 | 45.04 | 116.19 | 0.92 | 1.03 | 0.97 | 0.97 |
| 1903 | IMG_0027.JPG (red) | 562 | 88.08 | 27.87 | 25.67 | 125.01 | 0.91 | 1.09 | 0.92 | 0.95 |
| 1904 | IMG_0027.JPG (red) | 5248 | 267.87 | 82.03 | 81.46 | 176.22 | 0.92 | 1.01 | 0.99 | 0.98 |
| 1905 | IMG_0027.JPG (red) | 5241 | 270.01 | 83.02 | 80.38 | 153.84 | 0.90 | 1.03 | 0.97 | 0.98 |
| 1906 | IMG_0027.JPG (red) | 4206 | 242.31 | 74.35 | 72.03 | 143.74 | 0.90 | 1.03 | 0.97 | 0.98 |

Summary

File Edit Font

| Slice | Count | Total Area | Average Size | %Area | Perim. | Major | Minor | Angle |
|---|---|---|---|---|---|---|---|---|
| IMG_0002.JPG (red) | 39 | 103877.000 | 2663.513 | 0.487 | 183.401 | 56.738 | 54.603 | 113.1 |
| IMG_0003.JPG (red) | 39 | 100862.000 | 2586.205 | 0.473 | 174.150 | 53.901 | 51.784 | 95.91 |
| IMG_0004.JPG (red) | 90 | 241580.000 | 2684.222 | 1.134 | 183.401 | 56.648 | 54.417 | 88.50 |
| IMG_0005.JPG (red) | 81 | 232345.000 | 2868.457 | 1.090 | 192.002 | 59.505 | 57.009 | 93.47 |
| IMG_0006.JPG (red) | 96 | 235981.000 | 2458.135 | 1.107 | 172.151 | 53.528 | 50.721 | 82.71 |

# December 2021 macro without annotation.

To save the macro, open a new text window in FIJI (**File>New>Text Window**), copy and paste the following text, and save the file with an ijm extension, e.g., SERC_count_perkinsus.ijm.

```
/*
SERC_count_perkinsus.ijm
This FIJI/ImageJ macro is optimized for counting samples of RFTM incubated, NaOH digested, Lugol's stained Perkinsus samples that have been imaged with high resolution color micrographs. Dimensional parameters are in "pixel" units

Author: Teresa Vaillancourt, Ph.D. svmehitabel at gmail dot com
December 2021
License: BSD3

 Copyright 2021 Teresa Vaillancourt

 Redistribution and use in source and binary forms, with or without modification, are permitted provided that the following conditions are met:
 1. Redistributions of source code must retain the above copyright notice, this list of conditions and the following disclaimer.
 2. Redistributions in binary form must reproduce the above copyright notice, this list of conditions and the following disclaimer in the documentation and/or other materials provided with the distribution.
 3. Neither the name of the copyright holder nor the names of its contributors may be used to endorse or promote products derived from this software without specific prior written permission.
 THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

*/

FileName = getInfo("image.filename");
run("Duplicate...", "title=Copy1");+6
run("Duplicate...", "title=Copy2");
selectWindow(FileName);
run("Split Channels");
close();
close();
run("Auto Local Threshold", "method=Phansalkar radius=100 parameter_1=0 parameter_2=0 white");
run("Gaussian Blur...", "sigma=3");
run("Convert to Mask");
run("Watershed");
run("Set Measurements...", "area perimeter fit shape display redirect=None decimal=2");
run("Analyze Particles...", "size=50-1000000 circularity=0.85-1.00 show=[Overlay Masks] display exclude include summarize");
run("Labels...", "color=black font=12 show bold");
run("Magenta");
run("Flatten");
run("Copy");
close();
close();
selectWindow("Copy2");
run("Select All");
setBackgroundColor(0,0,0);
run("Clear", "slice");
run("Paste Control...");
run("Paste");
selectWindow("Copy1");
run("Images to Stack", "name=Stack title=[] use");
run("Make Montage...", "columns=2 rows=1 scale=1");
rename(FileName+"Montage");
```

To analyze a single image, first close any previously opened images (**File>Close All**), then open the image of interest and select **“Run”** in the text editor window containing the macro code.

To run the macro in batch mode, open the batch processor (**Process>Batch>Macro**), replace any existing content in the code window with the macro code above, select the directory containing image files to be analyzed and the directory to receive the analyzed montage files, then press **Process**. Note that the **Results** and **Summary** tables must be saved manually, if so desired.